SageTEX = Sage + TEX

Jason Grout
Drake University
jason.grout@drake.edu

What is Sage?

Sage: Mission Statement

Create a viable free open source alternative to Magma, Maple, Mathematica and Matlab.

Advantages of Sage

- Free
- Open-source
- Web notebook
- Industry-standard Python Language
- Integrates many standard open-source packages
- Interfaces to many commercial packages

Online Notebook

- Access Sage totally via the web
  - standard web browser
  - any operating system
  - even cell phones!
- *Nothing* to install
- One click to collaborate and share worksheets
- Typesetting, 2d graphics, interactive 3d graphics
- Buttons, sliders, etc., to explore problems

SageTEX

TEX document:

```
The number 2010 factors into $\sage{factor(2010)}$.
```

PDF Output:
The number 2010 factors into $2 \cdot 3 \cdot 5 \cdot 67$.

TEX document:

```
\sageplot{plot(-x^3+7*x,(x,-5,5))}
```

PDF Output:



How does it work?

1. Make TEX see the `sagetex.sty` file (e.g., copy it to your TEX file directory)
2. `pdflatex example.tex` – Makes an `example.sage` file containing the Sage source in your document
3. `sage example.sage` – Runs the Sage code and creates output for inclusion in the document
4. `pdflatex example.tex` – Inserts results of Sage code in PDF

Other ways to use SageTEX

- An option to include the Sage output with the tex file, to "freeze" things so people don't need Sage to TEX your file
- An option to use a remote Sage server to do the computations, so you don't have to have Sage installed locally
- Use SageTEX as a scripting language for TEX

Write explanations

```
\begin{sagesilent}
f(x,y)=x*sin(y); grad_f=f.gradient()
\end{sagesilent}
Let $f(x,y)=\sage{f(x,y)}$. Then $\nabla f=\sage{grad_f(x,y)}$.
\sageplot{plot_vector_field(grad_f, (x,-3,3), (y,0,3)),
          frame=True, aspect_ratio=1}
```

Let $f(x,y) = x\sin(y)$. Then $\nabla f = (\sin(y), x\cos(y))$.



Write questions

```
\begin{sagesilent}
m=identity_matrix(QQ,3)
m[0]=m[0]+m[1]
m[1]=m[1]-m[2]
m[2]=m[2]-2*m[1]
m[1]=m[1]+3*m[0]
m[0]=2*m[0]
\end{sagesilent}
Compute the rref of $\sage{m}$.
```

Compute the rref of $\begin{pmatrix} 2 & 2 & 0 \\ 3 & 4 & -1 \\ 0 & -2 & 3 \end{pmatrix}$.

Write answers

```
\begin{sagesilent}
M=random_matrix(QQ,3,4,algorithm='echelonizable',
    rank=3, upper_bound=10)
\end{sagesilent}
Compute the rref of $M=\sage{M}$.\\
Solution: $\mathrm{rref}(M)=\sage{M.rref()}$
```

Compute the rref of $M = \begin{pmatrix} 1 & 1 & -2 & -2 \\ 0 & 1 & -2 & -1 \\ 1 & 4 & -8 & -5 \end{pmatrix}$.

Solution: $\mathrm{rref}(M) = \begin{pmatrix} 1 & 0 & 0 & -1 \\ 0 & 1 & -2 & -1 \\ 0 & 0 & 0 & 0 \end{pmatrix}$

```
\begin{sagesilent}
f(x,y)=2*x^2*y+x*sec(y)+e^(-2*y)
resolution = 10
slope_field=plot_slope_field(-diff(f,x)/diff(f,y),
  (x,-3,3),(y,-2,3.5), plot_points=resolution)
phase=sum([implicit_plot(f(x,y)+i,(x,-3,3),(y,-2,3.5),
           plot_points=resolution*20, cmap='bone')
          for i in [-22..22,step=2]])
\end{sagesilent}

\sageplot{slope_field+phase, aspect_ratio=1,
  figsize=(5,5)}
```


3
2
1
-3
-2
-1
1
2
3
-1
-2

```
(8 pts) Solve the differential equation.  Show all work.  Y
  express the answer implicitly (i.e., you don't have to so
  $y$).

  \begin{equation*}
    \sage{diff(f(x,y),x)}
    + (\sage{diff(f(x,y),y)})\frac{dy}{dx}
    = 0
  \end{equation*}
```

(8 pts) Solve the differential equation. Show all work. You may express the answer implicitly (i.e., you don't have to solve for y).

$$4\,xy + \sec(y) + (x\tan(y)\sec(y) + 2\,x^2 - 2\,e^{(-2\,y)})\frac{dy}{dx} = 0$$

3d plots

```
\begin{sagesilent}
f(x,y)=x*sin(y)+y*cos(x)
\end{sagesilent}
\sageplot[width=2.4in]{plot3d(f,(x,-2,2),(y,-2,2))}
```



```
\begin{sagesilent}
mathematica('myplot=Plot3D[x^2-y^2,{x,-1,1},{y,-1,1}]')
mathematica('Export["%s/graphicsfile.png", myplot]'\
 %os.getcwd())
\end{sagesilent}
\includegraphics[width=3in]{graphicsfile}
```



```
\begin{sagesilent}
  G=graphs.PetersenGraph(); P=G.coloring()
\end{sagesilent}

A famous graph can be colored with
$\sage{G.chromatic_number()}$ colors:

\sageplot[width=1in]{G.plot(partition=P)}
```

A famous graph can be colored with 3 colors:



```
\begin{sagesilent}
  g = graphs.PetersenGraph()
  g.set_latex_options(tkz_style = 'Shade')
\end{sagesilent}
\sage{g}
```



Uses yet-to-be-released options

```
\begin{sagesilent}
H=graphs.HeawoodGraph()
H.set_latex_options(
 graphic_size=(4,4),
 vertex_size=0.2,
 edge_thickness=0.04,
 edge_color='green',
 vertex_color='green',
 vertex_label_color='red')
\end{sagesilent}

\sage{H}
```



Uses yet-to-be-released options

A Cayley Graph

```
\begin{sagesilent}
G=CyclicPermutationGroup(4)
C=G.cayley_graph(generators=[G((1,2,3,4)), G((1,4,3,2))])
C.set_pos(C.layout_circular())
C.set_latex_options(graphic_size=(8,6),
vertex_shape="rectangle",
edge_labels=True)
\end{sagesilent}
\sage{C}
```

A Cayley Graph



Uses yet-to-be-released options

Generate TEX using Sage

```
\begin{sagesilent}
var('x,y,n')
funcs=[x^2, x^n, sin(x),cos(x)]
table=r"\begin{table}  \centering  \begin{tabular}{cc}"
table+=r"$f(x)$ & $f'(x)$ \\\hline"
for f in funcs:
    table+=r"$%s$ & $%s$ \\"%(latex(f), latex(diff(f,x)))
table+=r"\end{tabular}\end{table}"
\end{sagesilent}
\sagestr{table}
```

$f(x)$	$f'(x)$
x^2	$2x$
x^n	$nx^{(n-1)}$
$\sin(x)$	$\cos(x)$
$\cos(x)$	$-\sin(x)$

```
\begin{sagesilent}
t=Tachyon(camera_center=(0,-4,1), xres = 800, yres = 600,
    raydepth = 12, aspectratio=.75, antialiasing = True)
t.light((0.02,0.012,0.001), 0.01, (1,0,0))
t.light((0,0,10), 0.01, (0,0,1))
t.texture('s', color=(.8,1,1), opacity=.9, specular=.95,
    diffuse=.3, ambient=0.05)
t.texture('p', color=(0,0,1), opacity=1, specular=.2,
    texfunc=1)
t.sphere((-1,-.57735,-0.7071),1,'s')
t.sphere((1,-.57735,-0.7071),1,'s')
t.sphere((0,1.15465,-0.7071),1,'s')
t.sphere((0,0,0.9259),1,'s')
t.plane((0,0,-1.9259),(0,0,1),'p')
\end{sagesilent}
\sageplot{t}
```



Credits and License

Special thanks to Dan Drake for the current SageTEX package.

Thanks also to Gonzalo Tornaria, Joe Wetherell, and Harald Schilly for previous versions of SageTEX. Thanks to Rob Beezer for the last three "fancy graph" examples.

Thanks!

Jason Grout
jason.grout@drake.edu

Sage: www.sagemath.org
SageTEX: on CTAN